A manageable support for the O.E.C.D. data on foreign trade by commodities

Bianchi, Carlo and Calzolari, Giorgio and Lischi, Pierluigi

IBM Scientific Center, Pisa, Italy

January 1978

IBM Italy
PISA SCIENTIFIC CENTER
G513 - 3567
January 1978

A MANAGEABLE SUPPORT FOR THE O.E.C.D. DATA ON FOREIGN TRADE BY COMMODITIES

Carlo Bianchi — IBM Pisa Scientific Center
Giorgio Calzolari — IBM Pisa Scientific Center
Pierluigi Lischi — during an EEC-NVI fellowship at the IBM Pisa Scientific Center

IBM Scientific Center
Via S. Maria 67
56100 Pisa, Italy

Abstract

This report describes the content and the format of five magnetic tapes in which the large volume of data on foreign trade by commodities distributed by O.E.C.D. has been condensed at the IBM Scientific Center of Pisa.

1. INTRODUCTION

The Directorate-General for Economic and Financial Affairs of the Commission of the European Communities and the IBM Scientific Center of Pisa have been carrying on a research on bilateral linkage of national econometric models.

Some preliminary results were presented at the 2nd International Conference on Dynamic modelling and Control of National Economies, Vienna, January 1977 [2] and, more recently, at the IBM UK Scientific Centre / IIASA Seminar on Models for Regional Planning and Policy Making, Laxenburg, Vienna, September 1977 [5].

The study has been mainly based on Barten's Import Allocation Model [1], where the bilateral trade flow of commodities between two countries is explained as a function of total imports of the importing country and of relative prices. For the computation of the relative prices on bilateral basis, however, official statistics were not available. For this reason it was decided to build the series of relative prices using the official data on bilateral trade by commodities supplied by the Organisation for Economic Co-operation and Development (O.E.C.D.) [4].

The official O.E.C.D. data for the period 1963-1975 were

supplied on 126 magnetic tapes; the first empirical problem to be solved was, therefore, that of transferring data on a more manageable support. The outcome of this preliminery work was the construction of five high density magnetic tapes, the description of which is the subject of this note. To have, however, more complete information on the characteristics of the data, reference should be made to [4].

2. TAPES ORGANIZATION

The characteristics of the five foreign trade by commodities tapes are the following:

- 9 tracks, phase encoding.
- Density is 6250 b.p.i.
- Record format is fixed blocked.
- Logical record length is 44 8-bit characters.
- Block size is 32560 characters.
- Labels are non-standard.
- Tape identifiers are the following:
  IMP63-69; IMP70-75; EXP63-68; EXP69-72; EXP73-75.

A tape contains more files concerning the imports or exports data of foreign trade by commodities for reporting countries vis-a-vis partner countries. A data file contains yearly trade of commodities for the O.E.C.D. reporting countries. Each data file is preceded by a label identifying each year and the file contained (IMPORTS or EXPORTS). The data are reported on the five tapes for the period 1963-1975. Each label ends with a double tape mark. Each data file ends with a double tape mark.

Tape organization:



2.1. Label

A label is one record of 44 characters. The label identifying the imports of year 1970 contains the detailed note:

'IMPORTS YEAR 1970 OCDE DATA'

The label identifing the exports of year 1970 contains the detailed note:

'EXPORTS YEAR 1970 OCDE DATA'

A double tape mark is inserted between the label record and the data file.

2.2. Data file

Each logical record (44 characters) in the data file contains the following items:

- Reporting country code
- partner country code
- January-March
  - Quantity
  - Value
- January-June
  - Quantity
  - Value

-January-September
  -Quantity
  -Value
-January-December
  -Quantity
  -Value

-S.I.T.C. code
-Quantity unit first period
-Quantity unit second period
-Quantity unit third period
-Quantity unit fourth period
-Control code first period
-Control code second period
-Control code third period
-Control code fourth period

The reporting and partner country are classified according to the O.E.C.D. geographic nomenclature [4]. The reporting country code is the same on all tapes, both before and after 1970. The partner country code is equal to the reporting country nomenclature after 1970; before 1970 the partner country code is given in the O.E.C.D. geographic nomenclature for the period 1961-1969 [4].

The data are reported for the four periods January-March, January-June, January-September, January-December; for each period there are quantity and value of the bilateral trade.

Products are classified according to Standard International

Trade Classification (S.I.T.C.) Revised codes. The product nomenclature is a 5-digit decimal code described in details in Statistical Paper Series M, Number 34, published by United Nations [6]; for the meaning of the non-numerical codes see [4]. The tapes containing trade data from year 1963 to 1969 contains data on total trade as well as trade at the 1,2,3 and 4 digits S.I.T.C. levels. The total trade is classified by 5 dashes (-) and the total at the 1,2,3 and 4 digit S.I.T.C. are classified by 1,2,3 and 4 digits of S.I.T.C. code followed by dashes.

Data records are written in the following format:

Reporting country code: bits 1-16, characters 1-2.

On the IBM System/370 memory this field occupies a halfword and represents an integer number corresponding to the value of the geographic code.

Partner country code: bits 17-32, characters 3-4.

On the IBM System/370 memory this field occupies a halfword and represents an integer number corresponding to the value of the geographic code.

January-March period: characters 5-12.

QUANTITY: bits 33-64, characters 5-8.

On the IBM System/370 memory this field occupies a fullword and represents a floating point number corresponding to the first period quantity (3 months).

VALUE: bits 65-96, characters 9-12.

This field occupies a fullword and represents a floating point number corresponding to the first period value given in

thousands of US dollars.

January-June period: characters 13-20.

QUANTITY: bits 97-128, characters 13-16.

On the IBM System/370 memory this field occupies a fullword and represents a floating point number corresponding to the second period quantity (6 months).

VALUE: bits 129-160, characters 17-20.

This field occupies a fullword and represents a floating point number corresponding to the second period value given in thousands of US dollars.

January-September period: characters 21-28.

QUANTITY: bits 161-192, characters 21-24.

On the IBM System/370 memory this field occupies a fullword and represents a floating point number corresponding to the third period quantity (9 months).

VALUE: bits 193-224, characters 25-28.

This field occupies a fullword and represents a floating point number corresponding to the third period value given in thousands of US dollars.

January-December period: characters 29-36.

QUANTITY: bits 225-256, characters 29-32.

On the IBM System/370 memory this field occupies a fullword and represents a floating point number corresponding to the fourth period quantity (12 months).

VALUE: bits 257-288, characters 33-36.

This field occupies a fullword and represents a floating point number corresponding to the fourth period value given in thousands of US dollars.

S.I.T.C. code: bits 289-328, characters 37-41.

This field contains the product nomenclature. It is a five characters string written in EBCDIC code. Every character is a digit of the S.I.T.C. code.

QUANTITY UNIT: bits 329-344, characters 42-43.

This field contains the quantity unit for four periods:

bits 329-332: first period quantity unit

bits 333-336: second period quantity unit

bits 337-340: third period quantity unit

bits 341-344: fourth period quantity unit

The meaning of the 4-bits of quantity unit is displayed in the following table:

0001 Quantity unit not specified

0010 Metric Ton

0011 Cubic Meter

0100 Thousands of Kilowatt hours

0101 Thousands of Litres

0110 Thousands of Metres

0111 Number (units or heads)

1000 Thousands of Pairs

1001 Thousands of Square metres

CONTROL CODE: bits 345-352, characters 44.

This field contains the control code for four perids.

bits 345-346: control code first period

bits 347-348: control code second period

bits 349-350: control code third period

bits 351-352: control code fourth period

The meaning of the two bits of the control code is displayed in the following table:

01 Quantity and Value data are available although a zero value is possible.

10 Data is not available for either Quantity or Value

11 Quantity data is not available but Value data is available

00 The quantity units are not equal all over the four periods; therefore quantities for the single quarters cannot be computed.

2.3. Record layout













3. EXAMPLES AND PERFORMANCES

To create one logical record to be stored into the data file of one tape by means of a FORTRAN program, the following coding subroutine could be profitably used. The requirements in the calling program are displayed in the heading comment lines of the subroutine.

```
*************************************************************************
*   PROPRIETA' DELLA IBM ITALIA
*   CENTRO SCIENTIFICO DI PISA
*
*   IBM SYSTEM/370. TO BE CALLED BY A FORTRAN PROGRAM.
*   CALLING PROGRAM REQUIREMENTS:
*
*    DIMENSION BUFFER(11),SITCCD(5),QVALUE(8),IQUNIT(4),ICNTR(4),
*   ,IGEOGR(2)
*    INTEGER*2 IGEOGR
*    ..........
*    CALL OCDECD(BUFFER,IGEOGR,SITCCD,ICNTR,IQUNIT,QVALUE)
*    WRITE(NTAPE,1)(BUFFER(L),L=1,11)
*  1 FORMAT(11A4)
*    ..........
*    WHERE:
*    BUFFER(11) IS WRITE BUFFER AREA
*    SITCCD(5) ARE FULLWORDS, EACH OF WHICH CONTAINING ONE ALPHANUMERIC
*              DIGIT OF THE SITC CODE, FOLLOWED BY 3 BLANKS.
*    QVALUE(8) ARE SINGLE PRECISION FLOATING POINT NUMBERS,ALTERNATIVELY
*              QUANTITY AND VALUE, IN THE SAME ORDER AS IN A TAPE RECORD
*    IQUNIT(4) ARE FULLWORD BINARY INTEGERS, CONTAINING THE QUANTITY
*              UNIT CODES OF THE FOUR PERIODS
*    ICNTR(4) ARE FULLWORD BINARY INTEGERS, CONTAINING THE CONTROL CODES
*             OF THE FOUR PERIODS
*    IGEOGR(2) ARE TWO HALFWORD BINARY INTEGERS CONTAINING THE REPORTING
*              AND PARTNER COUNTRIES CODES
*************************************************************************
```

```
          CSECT
          ENTRY OCDECD
          USING *,15
OCDECD    STM   14,12,12(13)
          L     2,0(1)  R2 CONTAINS POINTER TO BUFFER
          L     3,4(1)
          MVC   0(4,2),0(3)  IGEOGR HAS BEEN COPIED
          L     3,20(1)
          MVC   4(32,2),0(3)  THE VECTOR QVALUE HAS BEEN COPIED
          L     3,16(1)  R3 CONTAINS POINTER TO THE VECTOR IQUNIT
          L     4,0(3)  R4 CONTAINS IQUNIT(1)
          SR    5,5
          OR    5,4
          SLL   5,4
          L     4,4(3)
          OR    5,4
          SLL   5,4
          L     4,8(3)
          OR    5,4
          SLL   5,4
          L     4,12(3)
          OR    5,4   R5 CONTAINS IN BIT FORM THE 4 ELEMENTS OF IQUNIT
          L     3,12(1)  R3 CONTAINS NOW POINTER TO ICNTR
          SLL   5,2
          L     4,0(3)
          OR    5,4
          SLL   5,2
          L     4,4(3)
          OR    5,4
          SLL   5,2
          L     4,8(3)
          OR    5,4
          SLL   5,2
          L     4,12(3)
          OR    5,4
          ST    5,40(2)  THE VECTORS ICNTR AND IQUNIT HAVE BEEN COPIED
          L     3,8(1)
          MVC   36(5,2),0(3)  THE VECTOR SITCCO HAS BEEN COPIED
          LM    14,12,12(13)
          BR    14
          END
```

To process the tapes by means of a FORTRAN program, the following decoding subroutine could be profitably used. The requirements in the calling program are displayed in the heading comment lines of the subroutine.

```
*********************************************************************
*  PROPRIETA' DELLA IBM ITALIA
*  CENTRO SCIENTIFICO DI PISA
*  IBM SYSTEM/370. TO BE CALLED BY A FORTRAN PROGRAM.
*  CALLING PROGRAM REQUIREMENTS:
*   DIMENSION BUFFER(11),SITCCD(5),QVALUE(8),IQOUT(4),ICNTR(4),ISITC(5)
*   ..........
*   READ(NTAPE,1)(BUFFER(L),L=1,11)
* 1 FORMAT(11A4)
*   CALL OCDERD(BUFFER,IREP,IPART,SITCCD,QVALUE,IQOUT,ICNTR,ISITC)
*   ..........
*   WHERE:
*   BUFFER(11) IS READ BUFFER AREA
*   SITCCD(5) ARE FULLWORDS, EACH OF WHICH CONTAINING ONE ALPHANUMERIC
*             DIGIT OF THE SITC CODE, FOLLOWED BY 3 BLANKS.
*   QVALUE(8) ARE SINGLE PRECISION FLOATING POINT NUMBERS,ALTERNATIVELY
*             QUANTITY AND VALUE, IN THE SAME ORDER AS IN A TAPE RECORD
*   IQOUT(4) ARE FULLWORD BINARY INTEGERS, CONTAINING THE QUANTITY UNIT
*            CODES OF THE FOUR PERIODS
*   ICNTR(4) ARE FULLWORD BINARY INTEGERS, CONTAINING THE CONTROL CODES
*            OF THE FOUR PERIODS
*   IREP, IPART ARE FULLWORD BINARY INTEGERS, CONTAINING THE REPORTING
*               AND PARTNER COUNTRIES CODES
*   ISITC(5) ARE FULLWORD BINARY INTEGERS, CONTAINING THE 5 DIGITS OF
*            THE SITC CODE, CONVERTED ACCORDING TO THE FOLLOWING TABLE:
*     ALPHANUMERIC   NUMERIC
*          0            0
*          1            1
*         ...          ...
*          9            9
*          -           10
*          A           11
*          B           12
*          C           13
*         ...          ...
*          P           24
*********************************************************************
         CSECT
         ENTRY OCDERD
         USING *,15
OCDERD   STM   14,12,12(13)
         L     2,0(1)  REG.2 CONTAINS POINTER TO BUFFER
         L     4,4(1)  REG.4 CONTAINS POINTER TO IREP
         MVC   0(2,4),ZERO
         MVC   2(2,4),0(2)
         L     4,8(1)  REG.4 CONTAINS POINTER TO IPART
         MVC   0(2,4),ZERO
         MVC   2(2,4),2(2)
         L     4,12(1)  REG.4 CONTAINS NOW POINTER TO SITCCD
         L     5,28(1)  REG.5 CONTAINS POINTER TO THE VECTOR ISITC
         MVC   SITC(5),36(2)
         TR    SITC(5),TAB-96
         SR    6,6
         MVC   1(3,4),BLANK
         MVC   0(1,4),36(2)
```

```
         IC    6,SITC
         ST    6,0(5)
         MVC   5(3,4),BLANK
         MVC   4(1,4),37(2)
         IC    6,SITC+1
         ST    6,4(5)
         MVC   9(3,4),BLANK
         MVC   8(1,4),38(2)
         IC    6,SITC+2
         ST    6,8(5)
         MVC   13(3,4),BLANK
         MVC   12(1,4),39(2)
         IC    6,SITC+3
         ST    6,12(5)
         MVC   17(3,4),BLANK
         MVC   16(1,4),40(2)
         IC    6,SITC+4
         ST    6,16(5)
*  THE 5 ELEMENTS OF SITCCD AND OF ISITC HAVE BEEN COPIED INTO THEIR
*  AREAS
         L     4,16(1)  REG.4 CONTAINS NOW POINTER TO THE VECTOR QVALUE
*  WHOSE 8 ELEMENTS ARE QUANTITY AND VALUE OF THE FIRST QUARTER, THEN
*  QUANTITY AND VALUE OF THE SECOND QUARTER, ETC.
         MVC   0(4,4),4(2)
         MVC   4(4,4),8(2)
         MVC   8(4,4),12(2)
         MVC   12(4,4),16(2)
         MVC   16(4,4),20(2)
         MVC   20(4,4),24(2)
         MVC   24(4,4),28(2)
         MVC   28(4,4),32(2)
         L     4,20(1)  REG.4 CONTAINS NOW POINTER TO THE VECTOR IQOUT
         L     3,40(2)
         LR    5,3
         SLL   5,8
         SRL   5,28
         ST    5,0(4)
         LR    5,3
         SLL   5,12
         SRL   5,28
         ST    5,4(4)
         LR    5,3
         SLL   5,16
         SRL   5,28
         ST    5,8(4)
         LR    5,3
         SLL   5,20
         SRL   5,28
         ST    5,12(4)
         L     4,24(1)  REG.4 CONTAINS NOW POINTER TO THE VECTOR ICNTR
         LR    5,3
         SLL   5,24
         SRL   5,30
         ST    5,0(4)
         LR    5,3
```

```
         SLL   5,26
         SRL   5,30
         ST    5,4(4)
         LR    5,3
         SLL   5,28
         SRL   5,30
         ST    5,8(4)
         LR    5,3
         SLL   5,30
         SRL   5,30
         ST    5,12(4)
         LM    14,12,12(13)
         BR    14
ZERO     DC    F'0'
BLANK    DC    CL4'    '
TAB      DC    XL1'0A'
         DS    96C
         DC    XL4'0B0C0D0E'
         DC    XL4'0F101112'
         DC    XL1'13'
         DS    7C
         DC    XL4'14151617'
         DC    XL2'1819'
         DC    XL1'1A'
         DS    24C
         DC    XL4'00010203'
         DC    XL4'04050607'
         DC    XL2'0809'
SITC     DS    5C
         END
```

Reading and decoding the tape IMP70-75 for a quite simple computation, such as the construction of tables of aggregated values (at any aggregation level) of the imported commodities for each reporting country, requires approximately half an hour of virtual CPU time on an IBM/370 model 168, under the operating system VM/370-CMS [3].

The other tapes require approximately the same time of CPU.

REFERENCES

[1] Barten,A.P., "An Import Allocation Model for the Common Market", Cahiers Economiques de Bruxelles, 50, 3-14, (1971).

[2] Bianchi,C., Calzolari,G., Ranuzzi,P., "Price Competitiveness in a Bilateral Linkage Model for the EEC Economies", presented at the Second International Conference on Dynamic Modelling and Control of National Economies, Vienna, (1977).

[3] IBM, "Virtual Machine Facility/370: Command Language Guide for General Users", GC20-1804, IBM, New York, (1972).

[4] O.E.C.D., "Tape Specification on Foreign Trade by Commodities Tapes", O.E.C.D., Paris.

[5] Ranuzzi,P., "Bilateral Import and Export Functions for Italy", IBM UK Scientific Centre / IIASA Seminar on Models for Regional Planning and Policy Making, Laxenburg, Vienna, (1977).

[6] United Nations, "Standard International Trade Classification", Revised, Statistical Paper Series M, n. 34, New York, (1961).